Qspice - Command Reference Guide by KSKelvin

KSKelvin Kelvin Leung

Created on 8-4-2023
Last update on 4-27-2024

Comment
(Shortcut “;”)

Comment

Qspice : Comment - 3 Type.qsch | Comment - Chinese.qsch

3 Type of Comment

Type #1 : double slash //

```
//Type 1:
//Text Comment with double slash
```

Type #2 : Shortcut semicolon ";"




```
Type 2:
Text Comment
  * Right Click Text Box > This is a text comment
  * OR Shortcut ";"
```

Type #3 : semicolon in directive

```
.tran 5/1K ;Type 3 : Text Comment with semicolon in directive
.plot V(sin)
```

Comment with Chinese Character

Qspice New Feature
01/30/2024 Unicode text is now supported for comments placed on schematics.




```
//繁體中文範例
//简体中文范例
```

繁體中文範例 (Traditional Chinese)
简体中文范例 (Simplified Chinese)

```
.tran 5/1K ;瞬態響應
.plot V(sin) ;繪劃波型
```

.four
THD Computation

.four THD Computation

- Syntax: .four FREQ [HARMONICS] [PERIODS] expr1 [expr2 [expr3 [...]]]
  - FREQ : fundamental frequency
  - HARMONICS : number of harmonics to compute (Default HARMONICS=9)
  - PERIODS : number of period to be used to compute THD
  - expr1 : expression (e.g. V(out), I(V1) etc...)

- [HARMONICS]
  - Number of harmonics to compute




```
.four 1k 19 v(sgl):
Magnitude of Fundamental(RMS): 0.450158
Harmonic Frequency    Magnitude   Phase
   1     1.000e+03    1.000e+00     0.00°
   2     2.000e+03    5.890e-06   -90.00°
   3     3.000e+03    3.333e-01    -0.00°
   4     4.000e+03    5.890e-06   -90.00°
   5     5.000e+03    2.000e-01    -0.01°
   6     6.000e+03    5.890e-06   -89.99°
   7     7.000e+03    1.429e-01    -0.01°
   8     8.000e+03    5.890e-06   -89.99°
   9     9.000e+03    1.111e-01    -0.01°
  10     1.000e+04    5.890e-06   -89.99°
  11     1.100e+04    9.091e-02    -0.02°
  12     1.200e+04    5.890e-06   -89.99°
  13     1.300e+04    7.692e-02    -0.02°
  14     1.400e+04    5.890e-06   -89.99°
  15     1.500e+04    6.667e-02    -0.02°
  16     1.600e+04    5.890e-06   -89.99°
  17     1.700e+04    5.882e-02    -0.02°
  18     1.800e+04    5.890e-06   -89.99°
  19     1.900e+04    5.263e-02    -0.03°
THD = 45.686%(48.3398%)
```

.four THD Computation

Qspice : THD - basic.qsch

- .four THD
  - [HARMONICS] and [PERIODS] is not necessary input parameters
  - Magnitude and phase are normalized component equal 1 and 0 degree
  - To compare FFT, FFT is to plot magnitude in RMS
    - Therefore, it requires to calculate with Magnitude * Magnitude of Fundamental (RMS) in .four to match calculation in FFT
    - In waveform viewer, right click, select FFT



rms of harmonic 1 (fundamental)





Calculate V(sgl),rms from .four into FFT chart
Harmonic 1 : Magnitude*RMS = 1.000e+0*0.45 = 0.45
Harmonic 2 : Magnitude*RMS = 2.239e-5*0.45 = 0.00001
Harmonic 3 : Magnitude*RMS = 3.333e-1*0.45 = 0.15
:
:

Compare Qspice and LTspice .four log

Qspice

```
.four 1k v(sgl):
 Magnitude of Fundamental(RMS): 0.450158
 Harmonic  Frequency     Magnitude     Phase
    1      1.000e+03     1.000e+00      0.00°
    2      2.000e+03     2.239e-05    -90.00°
    3      3.000e+03     3.333e-01     -0.00°
    4      4.000e+03     2.239e-05    -90.00°
    5      5.000e+03     2.000e-01     -0.01°
    6      6.000e+03     2.239e-05    -90.00°
    7      7.000e+03     1.429e-01     -0.01°
    8      8.000e+03     2.239e-05    -90.00°
    9      9.000e+03     1.111e-01     -0.02°
 THD = 42.8795%(48.337%)
```

Equivalent
(for order e-05, error between simulator can be expected)

LTspice

Fourier Component in LTspice .four is Peak
Qspice : Magnitude * RMS * sqrt(2) = 1*0.450158*sqrt(2) = 0.6366

```
Fourier components of V(sgl)
DC component:0.500001

Harmonic     Frequency     Fourier      Normalized    Phase       Normalized
 Number        [Hz]       Component     Component   [degree]    Phase [deg]
    1       1.000e+3      6.366e-1      1.000e+0      90.00°        0.00°
    2       2.000e+3      2.000e-6      3.142e-6       0.00°      -90.00°
    3       3.000e+3      2.122e-1      3.333e-1      90.00°        0.00°
    4       4.000e+3      2.000e-6      3.142e-6       0.00°      -90.00°
    5       5.000e+3      1.273e-1      2.000e-1      90.00°        0.00°
    6       6.000e+3      2.000e-6      3.142e-6       0.00°      -90.00°
    7       7.000e+3      9.095e-2      1.429e-1      90.00°        0.00°
    8       8.000e+3      2.000e-6      3.142e-6       0.00°      -90.00°
    9       9.000e+3      7.074e-2      1.111e-1      90.00°        0.00°
Partial Harmonic Distortion: 42.880402%
Total Harmonic Distortion:   48.342146%
```

.func
Function

.inc
Include File

Include File (.inc) : HELP > Schematic Capture > Simulator > Include File (.inc)

Qspice : INC demo.qsch

- Include File
  - To include a file to execute by simulator
  - This allow to simplify schematic for directive or reuse purpose

- Example
  - This example use .inc to include a file called QspiceScript.txt
  - This script can
    - Calculate .func
    - Define .plot
    - Calculate .meas



QspiceScript.txt

```
.func dVOUT() d(V(out))
.plot dVOUT()
.plot V(out) V(in)
.meas Vo_pk max V(out)
.meas tau find time when V(out)=0.63*Vo_pk
```



Output Window

```
.meas vo_pk max v(out):
        9.99976 (at Time=0.05)
.meas tau find time when v(out)=0.63*vo_pk:
       0.004674    0.004674
```

.libpath
Include a Directory into the Library File Search Path

.libpath Include a Directory into the Library File Search Path

- .libpath
  - Syntax : .libpath <directory>
  - Search path for library .lib and include .inc directive
  - Search path priority
    1. Absolute path in .lib and .include
    2. Current working directory
    3. .libpath directories
    4. Qspice installation directory
       - Normal Install Path : C:\Program Files\QSPICE
  - If .libpath is used, .lib or .inc must after .libpath in netlist
    - Therefore, recommend to use Ctrl-Enter method after .libpath to add .lib / .inc to ensure this sequence




```
.dc Vce 0 10 0.01 Ib list 0.01 0.1 0.4 1 2
.libpath C:\QspiceKSKelvin\Qspice Common Library
.lib "KSKelvin_standard (from ltwiki).bjt"
.plot Ic(Q1)
```



.meas
Measure Statements

Available Syntax for .meas

HELP > Simulator > Command Reference > Measure(.meas)

- Syntax: .meas NAME find EXPRESSION1 at EXPRESSION2
- Syntax: .meas NAME find EXPRESSION1 when EXPRESSION2=EXPRESSION3
- Syntax: .meas NAME find EXPRESSION1 when EXPRESSION2=EXPRESSION3 td=5n cross=10
- Syntax: .meas NAME find EXPRESSION1 when EXPRESSION2=EXPRESSION3 cross=last
- Syntax: .meas NAME deriv EXPRESSION1 at EXPRESSION2
- Syntax: .meas NAME trig EXPRESSION1=EXPRESSION2
- Syntax: .meas NAME targ EXPRESSION1=EXPRESSION2
- Syntax: .meas NAME trig EXPRESSION1=EXPRESSION2 targ EXPRESSION3=EXPRESSION4
- Syntax: .meas NAME trig EXPRESSION1=EXPRESSION2 rise=1 targ EXPRESSION1=EXPRESSION2 rise=11
- Syntax: .meas NAME avg|max|min|pp|rms|integ EXPRESSION1
- Syntax: .meas NAME avg|max|min|pp|rms|integ EXPRESSION1 from EXPRESSION2 to EXPRESSION3
- Syntax: .meas NAME avg|max|min|pp|rms|integ EXPRESSION1 trig EXPRESSION2=EXPRESSION3 targ EXPRESSION4=EXPRESSION5
- Syntax: .meas NAME four FREQ EXPRESSION [...]
- Syntax: .meas NAME fra FREQ INPUT OUTPUT [...]

.meas with param

Qspice : meas - param.qsch

- .meas with param



.ic V(out)=0

.tran 1

Incomplete syntax: Return result at max simulation time

Syntax with param: Return result at max simulation time

Return result at measurement time

```
.meas Vnothing V(out)
.meas Vout param V(out)
.meas Vsin param V(sin)
.meas Vout@0.15s find V(out) at 0.15
```

Output Window

```
.meas vnothing v(out):
       0.999955
.meas vout param v(out):
       0.999955
.meas vsin param v(sin):
       0.853553
.meas vout@0.15s find v(out) at 0.15:
       0.776868
```

(1) measurement time

0.15 measurement time

Example of .meas in .ac analysis

Qspice : meas ac demo 01.qsch




```
.param fn = 1K
.param wn = 2*pi*fn
.param z = 0.2
Laplace=wn^2/(s^2+2*z*wn*s+wn^2)

.ac dec 100 100 1e4
.plot V(out)
```

Support 6 types of measure
avg/ave : average
max : maximum
min : minimum
pp : peak to peak
rms : root mean square
integ : integral

```
.meas gainpeak max dB(V(out))
.meas phase@2kHz find phase(V(out)) WHEN frequency=2K
.meas freq@+5dB find frequency WHEN dB(V(out))=5 cross=2
```

```
.meas gainpeak max db(v(out)):
     (    8.13428, -0.350754)  (at Frequency=954.993)
.meas phase@2khz find phase(v(out)) when frequency=2k:
     (  -165.064,          0)
.meas freq@+5db find frequency when db(v(out))=5 cross=2:
     (    1150.32,          0)
```

gainpeak find maximum of V(out) in dB



phase@2kHz find phase
at frequency is 2kHz

.meas results with .ac directive

Qspice : meas ac representation.qsch

- .meas results with .ac directive
  - Result is complex number : (real, imag)






```
.ac dec 100 10 100K
.plot V(out)
```

```
.meas v_complex find V(out) when frequency=1000
.meas v_dB_complex find dB(V(out)) when frequency=1000
.meas v_mag find abs(V(out)) when frequency=1000
.meas v_mag_dB dB(abs(V(out))) when frequency=1000
.meas v_phase phase(V(out)) when frequency=1000
```

$V(out) = 0.0247045 - j0.155223$

$20\log_{10} V(out) = -16.0722 - j12.2729$

** it log a complex number!

$|V(out)| = 0.157177\ (in\ Volt)$

$20\log_{10}|V(out)| = -16.0722\ (in\ dB)$

$\angle V(out) = -80.9596°$

Output Window

```
.meas v_complex find v(out) when frequency=1000:
     ( 0.0247045, -0.155223)
.meas v_db_complex find db(v(out)) when frequency=1000:
     (  -16.0722,  -12.2729)
.meas v_mag find abs(v(out)) when frequency=1000:
     (  0.157177,          0)
.meas v_mag_db db(abs(v(out))) when frequency=1000:
     (  -16.0722,          0)
.meas v_phase phase(v(out)) when frequency=1000:
     (  -80.9569,          0)
```

Example of .meas in .ac analysis for Q-factor

Qspice : meas - Q of LCR Resonant.qsch




```
.param L=10µ
.param C=1µ
.param R=0.2

.ac dec 1000 1 1G
.options LISTPARAM
.plot V(out)
```

[1] Add this option to display parameter evaluations result in output window

Series RLC:

$$Q = \frac{1}{R}\sqrt{\frac{L}{C}}$$



Q formula of Series RLC

```
.param Qcal 1/R*(L/C)**0.5
```

Q calculation from Bandwidth (BW) and Center Frequency (fo) : Q=fo/BW

```
.meas Vmax max mag(V(out))
.meas fo FIND frequency WHEN mag(V(out))=Vmax
.meas fL FIND frequency WHEN mag(V(out))=Vmax/sqrt(2) rise=1
.meas fH FIND frequency WHEN mag(V(out))=Vmax/sqrt(2) fall=last
.meas BW fH-fL
.meas Q fo/BW
```

Output Window

```
   --- Parameter Evaluations ---
TEMP        = 27        "CKTTEMP"
L           = 10µ       "10µ"
C           = 1µ        "1µ"
R           = 200M      "0.2"
QCAL        = 15.8114   "1/R*(L/C)**0.5"

C:\QspiceKSKelvin\01 User Guide and Script\01 Qspice Refer

Total elapsed time: 0.129028 seconds.
```

In simulation, it has .param calculation results

Simulation | Post Process

Output Window

```
.meas vmax max mag(v(out)):
     (  0.999835,          0)  (at Frequency=50350.1)
.meas fo find frequency when mag(v(out))=vmax:
     (   50350.1,          0)
.meas fl find frequency when mag(v(out))=vmax/sqrt(2) rise=1:
     (   48762.3,          0)
.meas fh find frequency when mag(v(out))=vmax/sqrt(2) fall=last:
     (   51946.9,          0)
.meas bw fh-fl:
     (   3184.61,          0)
.meas q fo/bw:
     (   15.8104,          0)
```

In post process, it has .meas calculation results

Simulation | Post Process

Example of .meas in .ac analysis

Qspice : meas tran demo 01.qsch




```
pulse 0 10 0 10n 10n 25µ 50µ

.tran 20m
.options MAXSTEP=0.1µ
```

Trick : If at=1m is removed, it return measurement at Tstop

```
.mease out@1ms V(out) at=1m

.mease out_steady avg V(out) from 19m to 20m
.mease Tr find Time when V(out)=0.9*out_steady

.meas Pin avg V(in)*-I(V1) from 19m to 20m
.meas Pout avg V(out)*I(R1) from 19m to 20m
.meas %error (Pin-Pout)/Pout*100
```

```
.mease out@1ms v(out) at=1m:
        6.74785       0.001
.mease out_steady avg v(out) from 19m to 20m:
        5.00264
.mease tr find time when v(out)=0.9*out_steady:
     0.000562296 0.000562296
.meas pin avg v(in)*-i(v1) from 19m to 20m:
        5.00497
.meas pout avg v(out)*i(r1) from 19m to 20m:
        5.00528
.meas %error (pin-pout)/pout*100:
     -0.0061396       0.02
```





.meas with trig and targ for time different calculation

Qspice : meas tran demo 02.qsch



.meas a2 trig V(signal)=0.8 cross=2
[time different from 2nd crossing V(signal)=0.8 to end of simulation]

.meas a1 trig V(signal)=0.5
[time different from V(signal)=0.5 to end of simulation]

.meas c1 trig V(signal)=-0.4 rise=1 targ V(signal)=-0.8 fall=2
[time different from 1st rising of V(signal)=-0.4 to 2nd falling of V(signal)=-0.8]

.meas with four (fourier component)

Qspice : meas fourier demo 01.qsch

```
.func rms2pk(in) in*sqrt(2)

sgl
V1
sin 0 rms2pk(1) 1K
V2
sin 0 rms2pk(0.6) 3K
V3
sin 0 rms2pk(0.2) 5K

.tran 5/1K
.plot V(sgl)
```

THD Total Harmonic Distortion

```
.four 1K V(sgl)
```

Fourier component with .meas

```
.meas xx four 1K V(sgl)
.meas |xx| abs(xx)
```

THD (.four)

```
.four 1k v(sgl):
 Magnitude of Fundamental(RMS): 0.999922
 Harmonic  Frequency     Magnitude    Phase
    1      1.000e+03     1.000e+00     0.00°
    2      2.000e+03     3.498e-08   105.41°
    3      3.000e+03     5.996e-01     0.00°
    4      4.000e+03     1.713e-08    64.78°
    5      5.000e+03     1.996e-01    -0.00°
    6      6.000e+03     1.341e-08    35.02°
    7      7.000e+03     1.292e-06    -8.50°
    8      8.000e+03     2.028e-08   -30.72°
    9      9.000e+03     2.366e-07    -5.73°
 THD = 63.1981%(63.1981%)
```

.meas with four

```
.meas xx four 1k v(sgl):
     (1.54886e-07, -0.999999)
.meas |xx| abs(xx):
       0.999999       0.005
```

Fourier component is a complex number (re+j*im)

Magnitude (rms) can be calculated with abs()



.meas with four (fourier component) [also with .step]

Qspice : meas fourier demo 02.qsch




```
.step dec param Vamp 100m 10 2
.tran 2m
.plot V(X)
```

THD Total Harmonic Distortion
.four 1K V(X)
Fourier component with .meas
.meas xx four 1K V(x) format : complex number
.meas |xx| abs(xx) convert complex to magnitude

```
.meas xx four 1k v(x):
   0 (-7.73763e-08,-0.0707086)
   1 (-2.48947e-07, -0.223511)
   2 (-3.44697e-06, -0.551818)
   3 (-7.78738e-06, -0.698485)
   4 ( 2.254e-06, -0.797766)
.meas |xx| abs(xx):
   0  0.0707086       0.002
   1   0.223511       0.002
   2   0.551818       0.002
   3   0.698485       0.002
   4   0.797766       0.002
```



Plot .meas data in Waveform Viewer

Qspice : meas waveform viewer.qsch

Method #1 :

[1] Add ".meas plot [Name]"

Method #2 :

[1] Run Simulation

[2] In Output Window

Right click in Post Process > Plot these Measurements



[3] X-axis default is .step parameter

[4] If you want to display X-axis parameter name

Right click x-axis > add bracket (or curly bracket) to parameter



.meas – FRA : fourier component between OUTPUT and INPUT

Qspice : meas - fra demo 01.qsch

- Syntax : .meas NAME **fra** FREQ INPUT OUTPUT [... range limits ...]
  - FRA : Fourier component of OUT at FREQ divided by the Fourier component of IN at FREQ
  - Range limits can be set with from/to or trig/targ syntax
  - Normalization is to the time domain RMS





.meas – FRA : fourier component between OUTPUT and INPUT

Qspice : meas - fra demo 02.qsch

- Frequency response (bode plot) from time domain with use of FRA
  - In this example, .step is used to sweep FRA frequency
  - Time domain simulation is performed at each FRA frequency



fmeas sweep from 100Hz to 10K with 50 points per decade



After .meas is ready in output window
Right click → Plot these Measurements



.model
Define Model

.model – ako Aliases (A Kind Of)

Qspice : model - ako.qsch

- ako (undocumented)
  - Aliases (**A K**ind **O**f)
  - Syntax : **ako:**
  - Modify parameters of an existing model
  - Example
    - Smdl_1 aliases model Smdl, but only changed Ron from 40 to 20




```
.plot V(x)/I(S2)
.plot V(x)/I(S1)

.tran 2
.model Smdl SW Ron=40 Roff=100 Vt=0.5 Vh=0
.model Smdl_1 ako: Smdl Ron=20
```



- Step model with ako!
  - With ako, it is possible to step a model in simulation
  - .step only accept numerical value
  - .model use numerical value for ako model name
  - Model name of device must be in curly bracket {}

{} must be used to step model




```
.dc V1 0 1 0.01
.plot I(D1)

.step param mdl list 4148 4007

.model 4148 ako:1N4148
.model 4007 ako:1N4007
```



.noise
Stochastic Noise Analysis

.noise Stochastic Noise Analysis

Qspice : noise - basic.qsch

- .noise
  - Stochastic Noise Analysis
  - .noise calculated results
    - Noise spectrum density per unit square root bandwidth ($\mathrm{V}/\sqrt{\mathrm{Hz}}$)
    - Input : Inoise
    - Output : Onoise
  - Notes
    - Noise analysis is performed without the needs of input voltage
    - In output window, it calculates total integrated output and input-referenced noise in rms
      - $\sqrt{\int (V_{noise})^2 \, df}$





Output Window (Simulation)

```
1 of 2 steps:  .step r=1000
Total integrated output-referenced noise: 12.8748µV rms
Total integrated input-referenced noise: 12.8748µV rms
2 of 2 steps:  .step r=2000
Total integrated output-referenced noise: 18.2077µV rms
Total integrated input-referenced noise: 18.2077µV rms
```

Output Window (Post-Processing)

```
.meas tmp integ v(onoise)*v(onoise):
   0 1.65761e-10
   1 3.31521e-10
.meas integrated sqrt(tmp):
   0 1.28748e-05      1e+07
   1 1.82077e-05      1e+07
```

.option / .options
Set Simulator Options

Set Simulator Options

Set Simulator Options

Syntax: .option NAME1=VALUE1 [NAME2=VALUE2 [...]]

Recognized Options

Name	Description	Default
ABSTOL	Absolute error tolerance	1e-12A
ACCT	Print accounting information	(not set)
ASCII	ASCII .qraw file	(not set)
BINARY	Override command line switch to use ASCII .qraw file	(not set)
BODEAMPFREQ	Frequency with the minimum perturbation amplitude. Set to 0. for constant amplitude.	(not set)
BODEINPUT[1]	Override input node for transfer function computation(AKA BODEIN)	auto
BODEPERIODS	Maximum number of periods to include in deconvolution	20
BODEREF	Reference node to use for Frequency Response Analysis	Node 0 (global ground)
BODEOUTPUT[1]	Override output node for transfer function computation((AKA BODEOUT)	auto
BODETOL	A Frequency Response Analysis relative tolerance	10.
CAPOP	0: Use model value 1: Use Meyer, >1 Use BSIM1	0
CHGTOL	Charge error tolerance	1e-14C
CSHUNT	Capacitance added from every node to ground(aka CMIN)	0F
DEFAD	Default MOSFET area of drain	$0m^2$
DEFAS	Default MOSFET area of source	$0m^2$
DEFL	Default MOSFET length	10µm
DEFW	Default MOSFET width	10µm
FEATHER	Trap integration damping factor	0
GMIN	Minimum conductance	1e-12℧
GMINSTEPS[2]	Number of Gmin steps	10
GSHUNT	Conductance added from every node to ground	0℧
ITL1	DC iteration limit	100
ITL2	DC transfer curve iteration limit	50
ITL4	Transient analysis iteration limit	10
KEEPOPINFO	Record operating point for small-signal analysis	(not set)
LAUNCHQUX[3]	Open the .qraw file in the waveform viewer after the simulation	(not set)
LIST[4]	Print an expanded netlist	(not set)
LISTPARAM	Print a list of the evaluated parameters	(not set)
MAXORD	Maximum integration order	2
MAXSTEP	Maximum timestep size for .bode and .tran	infinite
METHOD	Integration method(trap or Gear)	trapezoidal
MINBREAK[5]	Minimum time between breakpoints	0s
NOOPITER	Go directly to Gmin stepping	(not set)
NUMDGT	Number of significant digits in an ASCII .qraw file	15
PIVREL	Minimum relative matrix pivot	1e-3
PIVTOL	Minimum absolute matrix pivot	1e-13
RELTOL	Relative error tolerance	0.1%
RIC[6]	Impedance of source asserting initial conditions	1mΩ
SAVEPOWERS[7]	Compute and save the dissipation of components	(not set)
SEED[8]	Initialize the random number generater used in .param statements	
SEEDCLOCK	Initialize the random number generater with a 10Mhz clock and the process ID number(aka SEEDCLK).	(not set)
SRCSTEPS[2]	Number of source steps(aka ITL6)	10
TEMP	Operating temperature	27°C
TNOM	Nominal temperature(aka TREF)	27°C
TRTOL	Truncation error overestimation factor	2.5
TRTOL2	Another dimensionless truncation error guidance	1e-8
TRYTOCOMPACT	Try compaction for LTRA lines	(not set)
VNTOL	Voltage error tolerance	1µV

1] If a resistor is used to indicate where to insert the perturbation, the resistive divider's contribution is excluded.

2] Since an adaptive step size algorithms are used, the value of GMINSTEPS or SRCSTEPS is irrelevant unless set to zero, which means don't try the stepping algorithm.

3] Useful when running simulations from the command line. Don't use it if QSPICE64.exe or QSPICE80.exe are launched from the GUI.

4] Solely for internal diagnostic purposes. Probably not what you're looking for.

5] MINBREAK is automatically computed if left zero.

6] Inductor currents are asserted with the compliance of 1e9 * RIC.

7] Computes the true power dissipation while ignoring displacement currents. Implemented for BJTs, Capacitors, Diodes, Inductors, JFETs, MOSFET level 1, MOSFET level 2010 and VDMOS.

8] Used in .param functions Random() and Gauss(double sigma).

Simulator Option : CSHUNT and GSHUNT

Qspice : option - CSHUNT.qsch ; option - GSHUNT.qsch

- CSHUNT
  - Capacitance added from every node to ground (aka CMIN)
  - **Default CSHUNT=0F**
- Example to explain
  - Cshunt is equivalent to add Cshunt1/2/3 in node a/b/c





- GSHUNT
  - Conductance added from every node to ground
  - **Default GSHUNT=0℧**
- Example to explain
  - Gshunt is equivalent to add Rshunt1/2/3 = $\frac{1}{\text{GSHUNT}}$ in node a/b/c





Qspice : option - Gmin.qsch

- Gmin
  - Minimum conductivity that is added in parallel to every PN junction (diode, JFET, bipolars, MOSFET substrate diodes)
  - As double precision math isn’t accurate enough to find the bias point of two diode in series when reversed bias

Simulator Option : Gmin

Qspice : option - Gmin Diode.qsch

- Gmin
  - Minimum conductance
    - LTspice : Conductivity added to every PNjunction to aid convergence
  - Default Gmin=1e-12℧
- Explanation
  - Upper simulation use Gmin=1e-3 and measure I(V1) profile of D1a/D2a
  - Lower simulation force Gmin=0 (no effect of Gmin) and added R2a/R2b = $\frac{1}{1e-3}$, and measure I(V2) profile of D2a/D2b
  - This example demonstrate Gmin is equivalent to add shunt conductance for every PNjunction







Simulator Option : Gmin

Qspice : option - Gmin L (.dc).qsch ; option - Gmin L (.tran).qsch

- Gmin
  - Minimum conductance
  - In Qspice, Gmin also applied to inductor in .op and .tran initial inductor current calculation
    - Unlike PN junctions, gmin is only applied in inductor for its initial current calculation, but not added during transient analysis

.options Gmin=val



.step dec param val 1e-12 1e-4 10
.plot I(L1)
.op ← DC Operation Point Analysis



.options Gmin=val



.step dec param val 1e-12 1e-4 10
.plot I(L1)
.tran 1µ ← Transient Analysis I_L @ 0s
.meas Iinitial find I(L1) at 0



Simulator Option : MAXORD

- Maxord
  - Maximum integration order
  - **Default MAXORD=2**
  - MAXORD=1 forces Backward Euler, even less accurate than Gear

Simulator Option : KEEPOPINFO

- KeepOpInfo
  - Record operation point into .qraw for .ac and .tran
  - A file with extension .op.qraw is forced to create
    - This .op.qraw file contains DC operating point data
    - Open .op.qraw file with waveform viewer, a list of operating point is generated
  - .option keepopinfo is equivalent to use
    - Edit > Preferences
    - Click “Enable operating point display”
      - Hover device to display data only work with this enable







Simulator Option : Method – Trapezoidal or Gear

- Method
  - Integration method : Trapezoidal or Gear
  - **Default METHOD = Trapezoidal**
  - Trapezoidal : Area under every trapezoid is correct which can give the correct result
    - A disadvantage of using the trapezoidal method is that it can result in the occurrence of trapezoidal ringing when working with unrealistic circuit elements
    - In LTspice, it smooth trap ringing out but it is not in Qspice as one needs to know what the gates and flop truly see
  - Gear : Adds damping that isn't in the circuit and does not give as exact result, in general slower and less accurate compared to the trapezoidal method.
    - This is default method in Pspice

Simulator Option : Method – Trapezoidal or Gear

Reference by Mike Engelhardt in Qspice forum

- https://forum.qorvo.com/t/need-guidance-on-qspice-integration-method-and-this-feather-parameter/14393
- https://forum.qorvo.com/t/apparent-kirchhoff-s-law-violation/15048/2
- https://forum.qorvo.com/t/gear-vs-trap-what-are-those-and-how-to-optimize/16431
- https://ltwiki.org/files/LTspiceHelp.chm/html/integration_method_issues.htm

Engelhardt 26m

Aside from a few exceptions, Gear is both slower and less accurate than trap. The main problem with trap is a ringing artifact that bothers people, even though the area under each trapezoid is correct.

You've discovered the main point to offering both in the same simulator: if you get the same answer using both methods, you know that the solution is not affected by integration artifacts.

–Mike

Engelhardt Jul '23

FEATHER is an experimental parameter. It can be used to duplicate the de-tuned trap integration of HSPICE. It's use is not recommended.

As far as trap ringing, I realize it's disconcerting, but in a sense it's giving the right answer: The area under each trapezoid is correct, so one knows that the differential equations are correctly integrated. It can be reduced by either (i) using ".options method=Gear" (ii) stipulating a lower trtol, or (iii) stipulating a smaller maximum timestep. Gear is not recommended because it adds a substantial artificial damping to the circuit. I know of two cases where the use of Gear integration let an IC designer believe his circuit was stable until silicon said different and a turning Gear off confirmed.

Unlike some prior art, I don't smooth trap ringing out it out at all because one needs to know what the gates and flop truly see.

Engelhardt Aug '23

That looks like trapezoidal ringing - a situation where the numerically-integrated solution oscillates about the true solution timestep to timestep. A side effect is that current monitoring can look off[1].

In a sense, it's giving the correct answer, in that the area under each trapezoid has the correct area, but it is certainly disconcerting.

You can reduce or eliminate it by some of a combination of these methods:

1. Reduce trtol(.options trtol=1)
2. Reduce the maximum allow time step size(4th number on the .tran command)
3. Change to Gear integration.(.options method=Gear)

The last option is the most empathic way of getting rid of it, but it also introduces the greatest error. A simulation done with Gear integration can look stable when in fact, it is not. I've seen IC designers go to silicon only to learn they made the mistake of using Gear integration to check the IC's operation.

–Mike

1] Current monitoring is not really part of the solution of the circuit. It's the node voltages that are solved for, except for the voltage sources where the currents are part of the solution. Current monitoring is done as a forensic analysis of the circuit. Current reporting can be in error even if the rest of the solution is correct. I'll fix errors in report as they come up, e.g., today I fixed an issue in capacitor current reporting.

Simulator Option : Method – Trapezoidal Ringing

Qspice : option - Method (trap ringing).qsch

- Trapezoidal Ringing
  - Benefit of using the trapezoidal method is that the area under each trapezoid is calculated accurately, resulting in correct results
  - But trapezoidal ringing can occur when dealing with unrealistic circuit elements or function being integrated has sudden changes or contains high-frequency components
  - This example demonstrates how trapezoidal ringing occurs in a simulation when using a ramp source and derivative function
  - The first example illustrates how trapezoidal ringing occurs, while the second example forces the simulation step to the exact break point without any trapezoidal ringing




```
.option maxstep=1.94e-3
.tran 0 2
.plot V(dx)
.plot V(x)
```

In this setup, if you set different maxstep, you can get different magnitude of trap ringing






```
.option maxstep=1.94e-3
.tran 0 2
.plot V(dx)
.plot V(x)
```

Enable timectrl for Vsource Qspice simulation step can happen at exact break point (i.e. at 1s)



Simulator Option : Method – Trapezoidal Ringing

Qspice : option - Method (trap ringing).qsch

- Trapezoidal Ringing
  - Gearing introduces artificial damping that is not present in the circuit, effectively reducing ringing during integration
  - This approach is not generally recommended as it does not provide exact results





Simulator Option : RIC

Qspice : option - RIC L.qsch

- RIC
  - Impedance of source asserting initial conditions
  - Inductor currents are asserted with the compliance of 1e9 * RIC
  - Default RIC=1mΩ

- Important note
  - RIC only affect inductor current if .ic is used to define inductor initial current
  - In this simulation example, initial inductor current is plotted with Gmin=1e-12 and Gmin=1e3 with .ic I(L1)=1
    - When RIC=1e-3 (default), initial current is always equal .ic defined value

```
.options Gmin=1e-12
.options RIC=RIC
.step dec param RIC 1e-12 1e-3 10
```




```
.ic I(L1)=1
.plot I(L1)
.tran 1n
.meas Iinitial find I(L1) at 0
```




```
.options Gmin=1e3
.options RIC=RIC
.step dec param RIC 1e-12 1e-3 10
```




```
.ic I(L1)=1
.plot I(L1)
.tran 1n
.meas Iinitial find I(L1) at 0
```



Simulator Option : SAVEPOWERS

Qspice : option - SavePower (nmos).qsch

- Savepowers
  - Compute and save the dissipation of components
  - Computes the true power dissipation while ignoring displacement currents
    - Implemented for BJTs, Capacitors, Diodes, Inductors, JFETs, MOSFET level 1, MOSFET level 2010 and VDMOS
  - P() will be available in plot expression, and P() represent dissipation power of entire component
    - e.g. both gate and drain loss for nmos





- NMOS example
  - In this example, the NMOS is the only dissipation device, and its power is equal to the input power from V1 and V2
  - Input Power = V(G) * -I(V1) + V(D) * -I(V2)
  - Qspice calculated NMOS power = P(M1)
  - Average power of Input Power and P(M1) are equal (but Instantaneous power profile are different!!)
    - It can conclude that P(M1) include gate and drain power dissipation!

Simulator Option : SAVEPOWERS

Qspice : option - SavePower (nmos-gate).qsch

- Instantaneous Power
  - Example of gate power consumption. Instantaneous input power from V1 and dissipation power P(M1) is different, but their average power is the same
  - Gate model circuit R1 and C1 explains why instantaneous power profiles are different, as gate is not purely resistive but with reactive power in capacitance
  - Therefore, P(M1) in savepower option calculate instantaneous real power in device




```
.param fsw=5Meg
.option savepower
.tran 0 55/fsw 50/fsw 1/fsw/1e4
.plot V(x1)*-I(V1)
.plot P(M1)
.plot V(x1)
.plot V(x2)*-I(B1)
.plot V(x2,y2)*-I(B1)
.plot V(x2)
```

Same simulation

plots NMOS Gate or Gate modeling




```
.param fsw=5Meg
.option savepower
.tran 0 55/fsw 50/fsw 1/fsw/1e4
.plot V(x1)*-I(V1)
.plot P(M1)
.plot V(x1)
.plot V(x2)*-I(B1)
.plot V(x2,y2)*I(R1)
.plot V(x2)
```





Simulator Option : Seed and Seedclock

Qspice : option - Seed Seedclock.qsch

- Seed
  - Initialize the random number generator used in .param statements
  - Same random pattern is generated between Simulation Run





- Seedclock (aka Seedclk)
  - Initialize the random number generator with a 10Mhz clock and the process ID number(aka SEEDCLK)
  - Different random pattern is generated between Simulation Run





Simulator Option : TRTOL

- TRTOL
  - Truncation error overestimation factor
    - Qspice : TRTOL = 2.5
    - LTspice : TRTOL = 2
    - Other spice simulators : TRTOL = 7
  - CSHUNT, GSHUNT, GMIN, TRTOL

.param User-Defined Parameter

.param functions – random() and gauss(double sigma)

Qspice : param - random and gauss.qsch

- .param functions
  - Preparsing evaluation functions for .param
  - Random()
    - Random value between 0 and 1, uniform distribution
  - Gauss(σ)
    - Random number from Gaussian/Normal distribution with standard derivation σ and mean value at 0
  - Flat(x)
    - Random number between -x and x with uniform distribution
  - Mc(x,y)
    - Random number between x*(1+y) and x*(1-y) with uniform distribution









.step
Step User-Defined Parameter

.step with table : Batch Simulation

Qspice : Step with Table.qsch

- .step with table
  - By using table function, user can step integer N from 1, 2, 3, … and assign value to different parameters according to table
  - This is an example to sweep upper and lower resistor network with different resistance combination

```
.op
.step param N list 1 2 3
.param Ra table(N, 1,10K, 2,20K, 3,30K)
.param Rb table(N, 1,90K, 2,80K, 3,70K)
.plot V(x)/-I(V1)
.plot V(y)
```





N=1 Ra=10K Rb=90K

N=2 Ra=20K Rb=80K

N=3 Ra=30K Rb=70K

.tran
Non-Linear Transient Analysis

Syntax of .tran (Non-Linear Transient Analysis)

- Two syntax of .tran Non-Linear Transient Analysis
  - .tran TSTOP [UIC]
    - If MAXSTEP is required in this syntax, use *.options MAXSTEP* instead
  - .tran IGNORED TSTOP [TSTART [MAXSTEP]] [UIC]
    - Recommend to fill 0 at IGNORED. This syntax allows to set start recording time and maxstep
    - If need to limit .qraw file size, consider to specify Tstart and limit data to disk .qraw file size

1. Specify only the stop time

Syntax: .tran TSTOP [UIC]

Name	Description	Units
TSTOP	Total amount of time to simulate	s
UIC	Use initial conditions instead of solving for the initial bias point(SKIPBP)	

2. Traditional Berkeley Syntax

Syntax: .tran IGNORED TSTOP [TSTART [MAXSTEP]] [UIC]

Name	Description	Units
IGNORED[2]	An ignored value	s
TSTOP	Total amount of time to simulate	s
TSTART	Time to start recording waveform data to disk	s
MAXSTEP	Maximum time step size to allow	s
UIC	Use initial conditions instead of solving for the initial bias point(SKIPBP)	

.tran (Non-Linear Transient Analysis) : Syntax Examples

Only Stop Time Syntax




```
.tran 5   .options MAXSTEP=10µ
.tran 0 5 2 10µ
.plot V(sin)
```



Only Stop Time Syntax with .option MAXSTEP



Use .option to set MaxStep

```
.tran 5   .options MAXSTEP=10µ
.tran 0 5 2 10µ
.plot V(sin)
```



MaxStep increases simulation points

Traditional Berkeley Syntax




```
.tran 5   .options MAXSTEP=10µ
.tran 0 5 2 10µ
```

.tran Ignore Tstop Tstart MaxStep



.tran – UIC (Use Initial Condition)

Qspice : .tran - UIC.qsch

- UIC
  - **U**se **I**nitial **C**ondition
  - A DC operating point analysis (.op) is performed before starting the transient analysis (.tran). This directive suppresses this initialization
  - The node voltage is taken as zero if not specified
  - However, UIC is not a particularly recommended feature of SPICE (refer to UIC Help in LTspice), and reason is explained in next page
    - An alternative method for this example without UIC is to add initial condition directive
      .IC V(cap)=0









.tran – UIC (Use Initial Condition) and its Limitation

Qspice : .tran - UIC limitation.qsch

- UIC Limitation
  - Skipping the DC operating point analysis leads to nonphysical initial condition and may introduce difficulty in simulation
    - For example, voltage source in parallel to a capacitor required infinite current to charge in the first time step, which may return “time step too small” convergence fail

With UIC



.tran 100µ UIC
.option maxstep=100p
.plot I(C1)
.plot V(cap)



Without UIC



.tran 100µ
.option maxstep=100p
.plot I(C1)
.plot V(cap)



.tran – STARTUP (Not in Qspice)

LTspice : tran - startup.asc

- Startup
  - This is not a modifier directive in Qspice but in LTspice
    - It is needed for many of the switcher models in Ltspice according to Mike Engelhardt explanation
  - Startup modifier means independent source should be ramped on during the first 20us of the simulation
  - An equivalent approach in Qspice is to change voltage source from DC to ***PWL 0 0 20u [VDC]***









.tran (Non-Linear Transient Analysis) : .IC/IC Without UIC

Qspice : .tran - UIC and IC.qsch

Without UIC




```
.ic V(op)=2    .plot V(op)
.tran 10m
```



.IC Without UIC




```
.ic V(op)=2    .plot V(op)
.tran 10m
```



Instance IC W/o UIC




```
.ic V(op)=2    .plot V(op)
.tran 10m
```



.tran (Non-Linear Transient Analysis) : .IC/IC With UIC

Qspice : .tran - UIC and IC.qsch

With UIC





.IC With UIC





Instance IC With UIC





Batch mode Command

Qspice Execution Files

- Qspice execution files
  - Directory (default installation) : C:\Program Files\QSPICE
  - Schematic Capture and Waveform Viewer Program (HELP > Waveform Viewer)
    - Execution file : QUX.exe
    - Function #1 : Convert .qsch schematic to .cir
    - Function #2 : Export data from data file .qraw
  - QSPICE Simulator (HELP > Simulator)
    - Execution file : QSPICE64.exe [Enable Fast (less accurate) Math]
    - Execution file : QSPICE80.exe
    - Function : Run simulation from .cir
  - Post Processor (HELP > Post Processor)
    - Execution file : QPOST.exe
    - Function : Execute .meas and .four from .qraw

Batch command basic workflow

Qspice : Qspice_Batch_Command_LPFexample.bat / LPF Circuit.qsch

- Batch command workflow
  - Run CMD in Windows, in Command Prompt
  - Set path for Qspice program
    - path C:\Program Files\QSPICE\
  - Set variable name for working folder
    - set Qname=LPF Circuit
    - set Qpath=C:\QspiceKSKelvin\Qspice Batch
  - Goto schematic .qsch directory
    - cd %Qpath%
  - Convert .qsch to .cir (netlist)
    - QUX -Netlist "%Qname%.qsch"
  - Run Qspice simulation for .qraw
    - QSPICE64 -binary "%Qname%.cir"
    - QSPICE64 -ascii "%Qname%.cir" -r "%Qname%-ascii.qraw"
  - Export data from .qraw to .csv
    - QUX -Export "%Qname%.qraw" V(mid),V(out)
  - Post Process .meas and .four
    - QPOST "%Qname%.cir" -o "%Qname%.out"



QUX.exe : Netlist a Schematic (.qsch)

- Syntax for QUX buildtimestamp
  - QUX.exe -buildtimestamp

```
C:\Program Files\QSPICE>QUX.exe -buildtimestamp

C:\Program Files\QSPICE>Build Nov  3 2023 09:11:08
```

- Syntax for -Netlist
  - QUX.exe -Netlist <schematicfile> [-stdout]
    - <schematicfile> : name (+path) of a .qsch schematic, adds " " quotation for filename
    - If "-stdout" is not specified, the name of the netlist(.cir) file is computed from the name of the input .qsch file
    - [-stdout] : the netlist is printed on the console instead of to a file (not recommended since QSPICE employs a character set that most terminals can't handle

```
C:\QspiceKSKelvin\Qspice Batch>QUX -Netlist "%Qname%.qsch" -stdout

C:\QspiceKSKelvin\Qspice Batch>* LPF Circuit.qsch
L1 in mid 1
C1 mid 0 1
R1 out 0 1
V1 in 0 AC 1
L2 mid out 1
.ac dec 100 10K 1Meg
.plot V(mid) V(out)
.MEAS fc FIND frequency WHEN db(mag(V(out)))=-3
.end
```

QUX.exe : Export Datafile (.qraw)

- Syntax for -Export
  - QUX.exe -Export <datafile> <expr1[,expr2[,...]]> [Npoints] [CSV|SPICE|ASCII] [-stdout]
    - <datafile> : name of a .qraw file
    - <expr1[,expr2[,...]]> : expressions of data to extract
      - No space are allowed in the expression
      - Comma-separated expressions
    - [Npoints] : number of equally-spaced data points to extract
      - Default Npoints=1000
      - Npoints=1e308 or Npoints=all : all datapoints are extracted, waveform is not interpolated
    - [CSV|SPICE|ASCII]
      - CSV : Comma-Separated Value file
      - SPICE : .qraw in binary
      - ASCII : .qraw in ASCII
    - [-stdout] : extracted data is printed on the console instead of to a file
    - Example
      - QUX -Export "<filepath filename>" expr,expr2 all ascii ← no quotation mark is required for [Npoints] and [CSV|SPICE|ASCII]

QSPICE64.exe and QSPICE80.exe : QSPICE Simulator

- Syntax for output data .qraw name same as netlist .cir name
  - QSPICE64.exe -binary <netlistname> : Binary file format for output data .qraw
  - QSPICE64.exe -ascii <netlistname> : Ascii file format for output data .qraw
  - If 80 bit is used, change QSPICE64 to QSPICE80
- Syntax for specify output data .qraw name
  - QSPICE64.exe -[ascii/binary] <netlistname> -r <path> : specify the name of output data file
  - Example
    - set Qname=LPF Circuit
    - QSPICE64 -ascii "%Qname%.cir" -r "%Qname%-ascii.qraw"
- Syntax to directs the .qraw output to null (not saving a .qraw)
  - QSPICE64.exe <netlistname> -r NUL
    - This special usage is for user who write C++ dll datalogger and not prefer a .qraw to generate when simulating with batch mode

.qraw Binary Data format

Binary-binary.qraw

```
Title: * Binary.qsch
Date: Sun Nov  5 21:41:33 2023
Plotname: DC Transfer Characteristic
Flags: real
Abscissa: 1.000000000000000e+00 5.000000000000000e+00 lin
No. Variables: 4
No. Points: 3
Command: QSPICE64, Build Nov  3 2023 09:29:29
.param temp=27
Variables:
    0   V1  voltage
    1   V(a)    voltage
    2   I(V1)   current
    3   P(V1)   power
Binary:
NUL NUL NUL NUL NUL NUL ð? NUL NUL NUL NUL NUL NUL ð? NUL NUL NUL NUL NUL
```

Binary vs Ascii

Binary-binary.qraw | Binary-ascii.qraw

```
Title: * Binary.qsch
Date: Sun Nov  5 21:41:38 2023
Plotname: DC Transfer Characteristic
Flags: real
Abscissa: 1.000000000000000e+00 5.000000000000000e+00 lin
No. Variables: 4
No. Points: 3
Command: QSPICE64, Build Nov  3 2023 09:29:29
.param temp=27
Variables:
    0   V1  voltage
    1   V(a)    voltage
    2   I(V1)   current
    3   P(V1)   power
Values:
0       1.000000000000000e+00
    1.000000000000000e+00
    0.000000000000000e+00
    0.000000000000000e+00
1       3.000000000000000e+00
    3.000000000000000e+00
    0.000000000000000e+00
    0.000000000000000e+00
```

Newline </n>

```
00000130 76 6f 6c 74 61 67 65 0a 09 32 09 49 28 56 31 29 voltage..2.I(V1)
00000140 09 63 75 72 72 65 6e 74 0a 09 33 09 50 28 56 31 .current..3.P(V1
00000150 29 09 70 6f 77 65 72 0a 42 69 6e 61 72 79 3a 0a ).power.Binary:.
00000160 00 00 00 00 00 00 f0 3f 00 00 00 00 00 00 f0 3f .......ð?.......ð?
00000170 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
00000180 00 00 00 00 00 00 08 40 00 00 00 00 00 00 08 40 .......@.......@
00000190 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
000001a0 00 00 00 00 00 00 14 40 00 00 00 00 00 00 14 40 .......@.......@
000001b0 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 ................
```

Binary Format : Float 64

QPOST.exe : Post Processor to execute .meas and .four

- Syntax for Qpost.exe
  - Qpost <netlistname> -o <consoleoutput>
    - This will write .meas and .four results into a file for the console output
    - This result is equivalent in Qspice Post Process Output Window after Simulation is run




```
path C:\Program Files\QSPICE\

set Qname=resistor_network

set Qpath=C:\QspiceKSKelvin\01 User Guide and Script\02 Qspice Reference Guide\Batch Command\Qpost

cd %Qpath%

QUX -Netlist "%Qname%.qsch"

QSPICE64 -binary "%Qname%.cir"

QPOST "%Qname%.cir" -o "%Qname%.out"
```

resistor_network.out

```
.meas @amp param amp:
    0          1
    1        1.1
    2        1.2
    3        1.3
    4        1.4
    5        1.5
    6        1.6
    7        1.7
    8        1.8
    9        1.9
   10          2
.meas vrms rms v(in):
    0   0.707107
    1   0.777817
    2   0.848528
    3   0.919239
    4   0.989949
    5    1.06066
    6    1.13137
    7    1.20208
    8    1.27279
    9    1.3435
```

Curly Braces {}

Curly Braces { }

Qspice : Curly Braces with Random.qsch

- Curly Braces { }
  - In Pspice and Ltspice, the curly braces always meant evaluate the contents before proceeding to the simulation
  - But Qspice tries to figure out what can be solved before the simulation by itself and remove the necessary of using curly braces { }
    - Therefore, Qspice can call to use a parameter or a formula without curly braces { }
  - But user can still observe effect of curly braces in Qspice, an example is B-source with random(x) [with argument] function in B-source





** Qspice determines it is not a pre-process function and continues to run after warning is returned